
ANALYZ(COEF=(インプット パラメータ ベクトル), CONSTR, NAMES=(系列名のリスト),
PRINT, SILENT, TERSE, VCOV= 行列名)
方程式名のリスト ;

機能 :

ANALYZは、直近の OLSQ, LIML, LSQ, FIML, PROBIT, 等のプロシジャで推定されたパラメータの(非線形)関数セットの値とその共分散行列を計算します。さらに、関数セットが同時にゼロであるという仮説に対して Wald 検定値を計算します。もし、関数が線形なら、OLSQ コマンドの後で制約の F 検定と暗黙利に制約されたもとの係数がプリントされます。

計算方法は、推定されたパラメータ値の近傍で非線形関数を線形近似し、確率変数の線形関数の分散と共分散を計算する標準的な公式を用いて求めます。この「デルタ法」についての詳細は参考文献を参照して下さい。TSPは非線形関数に対する解析的微分を内部的に求め、ANALYZはVCOV行列とCOEFベクトルのサブセットを選択/再順序化するのに用い、これをHausman検定を行うのに用いることができます。

使用法:

ANALYZ の次に方程式名を続けます。線形モデル (OLSQ, INST, LIML, PROBIT) での推定プロシジャの後では、これらの方程式は係数に対応する変数名で表した計算したい推定された係数の関数を指定します。非線形モデル (LSQ, FIML) での推定プロシジャの後では、方程式は推定されたパラメータの関数を指定します。ANALYZは、別々に推定された推定式の分散を結合することはできません。これは、別々に推定された推定式のパラメータの共分散行列が計算されていないためです。なお、方程式はFRMLコマンドで事前に記述されていることが必要です。もし方程式の左辺に変数が定義されている場合は、パラメータから計算された値がその変数に代入されます。

もし(以前の OLSQ, INST, LIML, PROBIT 推定値から得られる右辺の変数名以外の) 系列変数が FRML に含まれる場合、添え字または SMPL(例えば、SMPL 10 10) によりスカラー変数を評価する必要があります。このようなFRMLに対する一つの応用は、推定されたパラメータや所得のようなデータに依存する弾力性の計算です。ANALYZは推定されたパラメータの共分散行列を用いて、さらにデータを固定定数と扱ってそのようなFRMLに対する標準誤差を計算します。

オプション:

COEF=は分析する方程式内のパラメータ値を含むベクトルです。このベクトルは NAMES= オプションにあるパラメータと、与えた VCOV 行列にも対応します。デフォルトは@COEFです。

CONSTR/NOCONSTR は、OLSQ に続いて ANALYZ を行ったとき、制約付の回帰係数の計算を行えるようにします。制約付の係数を計算したりプリントしないようにするためには、NOCONSを用います。

NAMES=は、パラメータ名のオプションリストを指定します。そのパラメータ名のラベルは対応する共分散行列はVCOV=オプションにより与えます。デフォルトは@RNMSです。

PRINT/NOPRINT は、ANALYZ コマンドのインプットの内容 (パラメータ値と方程式) を出力するかどうかを指示します。既定値はNOPRINTで、結果だけがプリントされます。

ANALYZ

SILENT/NOSILENTはアウトプットをしないようにします。結果は@RNMSA, @COEFA, 等の名前で保存されます。

TERSE/NOTERSEは、作成すべき「最小の」アウトプットを指定します。OLSQ の後の ANALYZ では、カイ2乗検定を省略しますが F 値はプリントします。さらに制約付の係数の出力も省略しますが、NOCONST が用いられていても (@COEFC 等の名前で) 保存されます。

VCOV=は、NAME=で指定されたインプット・パラメータの分散共分散行列の名前を与えます。これら2つのオプションを使うと、標準的な推定プロシジャで得られる@VCOV 行列以外の行列に ANALYZを行なえるようになります。デフォルトは@VCOVです。

例:

ラグ付従属変数のあるモデルで、「長期の」係数を計算します。

```
FRML LR1 ALPHAIR=ALPHA/(1-LAMBDA) ;
FRML LR2 PHILR=PHI/(1-PSI) ;
ANALYZ LR1 LR2 ;
```

トランスログ型システムにおける、制約付パラメータに対する標準誤差を計算するのに (EQSUBと共に) ANALYZを用いる例については、このレファレンス・マニュアルのEQSUB コマンドの項を参照して下さい。

次の例は、データ内の観測値と標本データの平均の両方で弾力性 (とその標準誤差) を計算する2つの方法です。

```
FRML EQ1 LQ1 = A1 + B1*LP1 + B2*LP2 + B12LP1*LP2 + B13*LP1*LP3 + B23*LP2*LP3;
FRML EL1 ELD1 = B1 + B12*LP2 + B13*LP3; ? d(LQ1)/d(LP1)
SMPL 48,95;
LSQ EQ1;
SMPL 72,72; ? LP2, LP3, etc. の1972年の値の弾力性を得る
ANALYZ EL1;
SMPL 48,95; ? LP2, LP3, etc.の標本平均値で弾力性を計算する
MSD LP2 LP3; ? 平均を固定した人工的な'1996'年の観測値を作成する
SMPL 96,96;
UNMAKE @MEAN LP2 LP3;
ANALYZ EL1;
```

これはOLSQの後でANALYZ コマンドを使う例です。これは2つの係数の和がゼロであるという仮説のカイ2乗検定値を計算します (この検定統計量は、標準的な F 統計量に等しくなります)。

```
OLSQ Y C X1 X2 ;
FRML SUM X1+X2 ;
ANALYZ SUM;
```

たとえば、任意の次数の多数のパラメータがあるシステムから2,3のパラメータとそのVCOV 行列だけを抽出したい場合には、次のようにします。

```
SUPRES COEF;
LSQ (SILENT) EQ1-EQ50; ? 多数の方程式の推定
SUPRES;
DOT B1-B5;
FRML EQ..=. ; ? FRML EQB1 B1=B1; etc. の作成
ENDDOT;
ANALYZ EQB1-EQB5; ? 後で使うために5つのパラメータだけをプリント
RENAME @VCOVA VCVB1_5;
RENAME @COEFA CB1_B5;
```

アウトプット:

もし PRINT オプションが指定されると, ANALYZ は, タイトル, 入力パラメータ名, 記号での方程式, 導出された関数とそれらの標準誤差, そして関数が同時にゼロであるという検定の χ^2 値の表をプリントします. この χ^2 値の自由度は方程式の本数になります. さらに, χ^2 検定値に対する P-値 (有意水準) も出力されます. プリント・オプションがオフ (既定値) なら, 導出された関数と χ^2 検定値だけがプリントします.

PRINT オプションの指定に拘わらず, ANALYZ もパラメータの計算値とその共分散行列を, それらが推定結果のようにデータ領域に保存します. 結果は次のような名前で検索できます.

名前	タイプ	長さ	変数の内容
@RNMSA	リスト	方程式数	計算されたパラメータの名前
@WALD	スカラー	1	ワルド検定の値
@NCOEFA	スカラー	1	計算されたパラメータの数
@NCIDA	スカラー	1	ワルド検定の自由度
%WALD	スカラー	1	ワルド検定の P-値 (有意点)
@COEFA	ベクトル	方程式数	計算されたパラメータの値
@SESA	ベクトル	方程式数	計算されたパラメータの標準偏差
@TA	ベクトル	方程式数	t-統計値 (漸近的正規)
@VCOVA	行列	方程式数*方程式数	計算されたパラメータの共分散行列

方法:

TSP での以前の推定が, K 個のパラメータベクトル $b = @COEF$ とそれらの分散行列 $var(b) = @VCOV$ が保存されており, M 個の関数 $f(b)$ に対する値と標準誤差がほしいと仮定します. ANALYZ は b に関する f の 1 次微分を解析的に得ます: $df/db = G$ ($K \times M$ 行列). 関数 $f(b)$ と行列 G は, $f(b)$ に現れるその他の定数あるいはデータ値とともに b の現在の値で評価されます. $f(b)$ に対する分散共分散は, したがって (漸近的に, あるいは $f(b)$ が線形なら厳密に) $var(f(b)) = G' @VCOV * G$ となります. これはデルタ法として知られています. たとえば, $M=1$ と $f(b) = f_1 = 2*b_1$ なら, $G=2$ ($K > 1$ ならばゼロ), そして $var(f_1) = 4*var(b_1)$ になります.

方程式が線形で OLSQ コマンドで推定すると, ANALYZ は 1 組の結合仮説 ($@FST = @WALD / @NCIDA$) に対する F 統計量をプリントします. さらに, ANALYZ は暗黙的に制約が課されたもとの係数とその標準誤差を計算し, プリントします. これらは, $@COEF$ と $@VCOVC$, 等の名前で保存されます.

参考文献:

Bishop, Y.M.M., S.E. Fienberg, and P.W. Holland, **Discrete Multivariate Analysis: Theory and Practice**, MIT Press, Cambridge, MA, 1975, pp.486-502.

Gallant, A. Ronald, and Dale Jorgenson, "Statistical Inference for a System of Simultaneous, Non-linear, Implicit Equations in the Context of Instrumental Variable Estimation," **Journal of Econometrics**, **11**, 1979, pp.275-302.

Gallant, A. Ronald, and Albert Holly, "Statistical Inference in an Implicit, Nonlinear, Simultaneous Equation Models in the Context of Maximum Likelihood Estimation," **Econometrica**, **48**, 1980, pp.697-720.